Heck, *Introduction to Maple*
Exercises
Chapter 2

1. Consider the following Maple session

   ```
   > 3^2:
   > 4^2;
   ```
   $$16$$
   ```
   > % + %%;
   ```

   does the last instruction make sense? If so, what is the result? if not, why?

   ```
   > 3^2:
   > 4^2;
   ```
   $$16$$
   ```
   > % + %%;
   ```
   $$25$$

   It makes sense because ***Maple*** does not need to have the result printed to screen in order to know the outcome of a command.

2. Explain the different results of the following ***Maple*** commands.

   (a) `x:y;`

   (b) `x/y;`

   (c) `x\y`

   ```
   > x:y;
   ```
   $$y$$

   Maple simply returns y. Putting the colon behind x tells ***Maple*** to execute that statement but not print any output.

   ```
   > x/y;
   ```
   $$\frac{x}{y}$$

   This gives the fraction x/y.

   ```
   > x\y;
   ```
   $$xy$$

   The backslash, \, is a continuation character.

3. In this exercise you can practice your skills in using the help system of ***Maple***.

(a) Suppose that you can want to select from an equation, **e.g.**, $1 = \cos(x)^2 + \sin(x)^2$, only the left or right side. How can you easily do this in ***Maple***?

```
> eqn := 1=cos(x)^2+sin(x)^2;
```

$$eqn := 1 = \cos(x)^2 + \sin(x)^2$$

```
> lhs(eqn);
```

$$1$$

```
> rhs(eqn);
```

$$\cos(x)^2 + \sin(x)^2$$

(b) Suppose that you want to compute the continued fraction approximation of the exponential function; can ***Maple*** do this for you? If yes, carry out the computation.

```
> ?confrac
> convert(exp(x),confrac,x);
```

$$1 + \cfrac{x}{1 + \cfrac{x}{-2 + \cfrac{x}{-3 + \cfrac{x}{2 + \frac{1}{5}\,x}}}}$$

(c) Suppose that you want to factor the polynomial $x^8 + x^6 + 10\,x^3 + 8\,x^2 + 2\,x + 8$ modulo 13. Can ***Maple*** do this? If yes, carry out this factorization.

```
> Factor(x^8+x^6+10*x^3+8*x^2+2*x+8)mod 13;
```

$$(x^2 + 8\,x + 9)\,(x + 7)\,(x^2 + 11\,x + 12)\,(x^3 + 6\,x + 4)$$

(d) Suppose that you want to determine all subsets of the set {1, 2, 3, 4, 5}. How can you do this in ***Maple***?

```
> with(combinat,powerset);
```

$$[powerset]$$

```
> powerset({1,2,3,4,5});
```

{{}, {1, 2, 3, 4, 5}, {2, 3, 4, 5}, {3, 4, 5}, {1, 3, 4, 5}, {4, 5}, {1, 4, 5}, {2, 4, 5},
{1, 2, 4, 5}, {5}, {1, 5}, {2, 5}, {1, 2, 5}, {3, 5}, {1, 3, 5}, {2, 3, 5}, {1, 2, 3, 5},
{1}, {2}, {1, 2}, {3}, {1, 3}, {2, 3}, {1, 2, 3}, {4}, {1, 4}, {2, 4}, {1, 2, 4},
{3, 4}, {1, 3, 4}, {2, 3, 4}, {1, 2, 3, 4}}

4. Load the `numtheory` package by entering `with(numtheory);` You may recognize some functions from number theory; some of the routines in this package are useful in answering the following questions.

```
> with(numtheory);
```

```
Warning, the protected name order has been redefined and unprotected
```

[*GIgcd*, *bigomega*, *cfrac*, *cfracpol*, *cyclotomic*, *divisors*, *factorEQ*, *factorset*, *fermat*, *imagunit*, *index*, *integral_basis*, *invcfrac*, *invphi*, *issqrfree*, *jacobi*, *kronecker*, λ, *legendre*, *mcombine*, *mersenne*, *minkowski*, *mipolys*, *mlog*, *mobius*, *mroot*, *msqrt*, *nearestp*, *nthconver*, *nthdenom*, *nthnumer*, *nthpow*, *order*, *pdexpand*, ϕ, π, *pprimroot*, *primroot*, *quadres*, *rootsunity*, *safeprime*, σ, *sq2factor*, *sum2sqr*, τ, *thue*]

(a) Build a list of all integers that divide 9,876,543,210,123,456,789.

```
> divisors(9876543210123456789);
```

{1, 3, 9, 13, 39, 117, 6353, 8969, 19059, 26907, 57177, 80721, 82589, 116597, 247767, 349791, 743301, 1049373, 2222222223, 253244697695473251, 759734093086419753, 172736296240157, 3292181070041152263, 518208888720471, 1554626666161413, 39862222209267, 6666666669, 119586666627801, 1097393690013717421, 13287407403089, 9411851848793, 9876543210123456789, 4444444443, 173333333277, 57777777759, 19259259253, 13333333329, 367062222102927, 1101186666308781, 84414899231824417, 28235555546379, 84706666639137, 122354074034309, 1481481481, 56980057, 170940171, 512820513, 740740741}

(b) Find the prime number that is closest to 9,876,543,210,123,456,789.

```
> nextprime(9876543210123456789);
```

$$9876543210123456803$$

```
> prevprime(9876543210123456789);
```

$$9876543210123456781$$

```
> %%-9876543210123456789;
```

$$14$$

```
> 9876543210123456789-%%;
```

$$8$$

The closest prime to 9,876,543,210,123,456,789 is 9,876,543,210,123,456,781.

(c) What is the prime factorization of $5^{(5^{(5^5)})}$?

```
> ifactor(5^(5^(5^5)));

Error, numeric exception: overflow
```

(d) Expand the base e of the natural logarithm as a continued fraction up to 10 levels deep.

```
> cfrac(exp(1),9);
```

$$2+\cfrac{1}{1+\cfrac{1}{2+\cfrac{1}{1+\cfrac{1}{1+\cfrac{1}{4+\cfrac{1}{1+\cfrac{1}{1+\cfrac{1}{6+\cfrac{1}{1+...}}}}}}}}}$$

5. In ***Maple***, what is the difference between $\frac{1}{3}+\frac{1}{3}+\frac{1}{3}$ and $\frac{1.0}{3.0}+\frac{1.0}{3.0}+\frac{1.0}{3.0}$?

```
> 1/3+1/3+1/3;
```

$$1$$

```
> 1.0/3.0+1.0/3.0+1.0/3.0;
```

$$.9999999999$$

How many digits of accuracy are necessary to make ***Maple*** think that they are the same, or will it ever consider them to be the same?

```
> Digits:=1000:
> 1.0/3.0+1.0/3.0+1.0/3.0;
```

```
.99999999999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999999999999\
99999999999999999999999999999999999999999999999999
```

6. Find the floating-point approximation of $(e)^{(\frac{\pi\sqrt{163}}{3})}$ using a precision of 10, 20, and 30 digits, respectively.

```
> restart;
> evalf(exp(Pi*sqrt(163)/3),10); #10 digit precision
```

$$640319.9998$$

```
> evalf(exp(Pi*sqrt(163)/3),20); # 20 digit precision
```

$$640320.00000000060543$$

```
> evalf(exp(Pi*sqrt(163)/3),30); # 30 digit precision
```

$$640320.000000000604863735049036$$

7. Calculate $\pi^{(\pi^\pi)}$ to nine decimal places.

First, we expect that we should simply execute

```
> evalf(Pi^(Pi^Pi),9);
```

```
> evalf(Pi^(Pi^Pi),9);
```

$$.134016366\,10^{19}$$

but this only gives 9 digits, **NOT** 9 decimal places. We need a few more digits to get 9 decimal places.

```
> evalf(Pi^(Pi^Pi),30);
```

$$.134016418300635743529744912966\,10^{19}$$

8. Compute this exercise in a floating-point precision of eight decimal places. What is the result of 310.0 320.0 330 - $\sqrt{310.0\,320.0}$ $\sqrt{320.0\,330.0}$ $\sqrt{330.0\,310.0}$?

```
> Digits:=8:
> 310.0*320.0*330 -
> sqrt(310.0*320.0)*sqrt(320.0*330.0)*sqrt(330.0*310.0);
```

$$-1.$$

```
> restart;
```

9. Do you remember which of the numbers $\frac{19}{6}$, $\frac{22}{7}$, and $\frac{25}{8}$ is a fairly good rational approximation of π? Use ***Maple*** to find the best of these three numbers. Find the best rational approximation $\frac{a}{b}$ of π, where a and b are natural numbers less than 1000 (Hint: look at the continued fraction expansion of π).

```
> with(numtheory):
```

```
Warning, the protected name order has been redefined and unprotected
```

```
> evalf(19/6);evalf(22/7);evalf(25/8);
```

$$3.166666667$$
$$3.142857143$$
$$3.125000000$$

```
> cfrac(Pi,3);
```

$$3 + \cfrac{1}{7 + \cfrac{1}{15 + \cfrac{1}{1 + ...}}}$$

```
> 3+1/(7+1/(15+1/(1)));
```

$$\frac{355}{113}$$

```
> evalf(%);
```

$$3.141592920$$

10. Check that $\sqrt{2\sqrt{19549}+286}$ is equal to $\sqrt{113}+\sqrt{173}$.

```
> sqrt(2*sqrt(19549)+286)-(sqrt(113)+sqrt(173));
```

$$0$$

```
> verify(sqrt(2*sqrt(19549)+286),sqrt(113)+sqrt(173));
```

$$\mathit{true}$$

```
> evalb(sqrt(2*sqrt(19549)+286)=sqrt(113)+sqrt(173));
```

$$\mathit{true}$$

```
> restart;
```

11. In ***Maple***, transform $\frac{1}{\sqrt{3}+1}$ into an expression of the form $a + b\sqrt{3}$, with rational numbers a and b.

```
> rationalize(1/(sqrt(3)+1));
```

$$-\frac{1}{2} + \frac{1}{2}\sqrt{3}$$

12. Let θ be a root of the polynomial $\theta^3 - \theta - 1$ and consider the extension of the field of rational numbers with θ. So, we consider expressions of the form $a + b\,\theta + c\,\theta^2$, where a, b, and c are rational numbers, and in calculations with these expressions we apply the identity $\theta^3 = \theta + 1$. Transform with ***Maple*** $\frac{1}{\theta^2+1}$ into an expression of the form $a + b\,\theta + c\,\theta^2$, where a, b, and c are rational numbers.

```
> alias(theta=RootOf(x^3-x-1)):
> simplify(1/(theta^2+1));
```

$$\frac{4}{5} - \frac{2}{5}\theta^2 + \frac{1}{5}\theta$$

13. Show that ***Maple*** knows that the exponential power of a complex number can be written in terms of cosine and sine of the real and imaginary parts of that number. Also calculate $e^{(\frac{\pi I}{12})}$ in that form.

```
> convert(exp(I*z),trig);
```

$$\cos(z) + I\sin(z)$$

```
> assume(x,real):assume(y,real):
> evalc(exp(x+I*y));
```

$$e^{x\tilde{}}\cos(y\tilde{}) + I\,e^{x\tilde{}}\sin(y\tilde{})$$

```
> evalc(exp(Pi*I/12));
```

$$\cos(\frac{1}{12}\,\pi) + I\sin(\frac{1}{12}\,\pi)$$

```
> convert(%,radical);
```

$$\frac{1}{4}\sqrt{2}\,(1+\sqrt{3}) + \frac{1}{4}\,I\,\sqrt{2}\,(\sqrt{3}-1)$$

14. Show with ***Maple*** that $\tanh(\frac{z}{2}) = \frac{\sinh(x)+I\sin(y)}{\cosh(x)+\cos(y)}$, for any complex number $z = x + y\,I$ with real x and y.

```
> restart:
> z:=x+I*y;
```

$$z := x + I\,y$$

```
> evalc(tanh(z/2));
```

$$\frac{\sinh(\frac{1}{2}\,x)\cosh(\frac{1}{2}\,x)}{\sinh(\frac{1}{2}\,x)^2 + \cos(\frac{1}{2}\,y)^2} + \frac{I\sin(\frac{1}{2}\,y)\cos(\frac{1}{2}\,y)}{\sinh(\frac{1}{2}\,x)^2 + \cos(\frac{1}{2}\,y)^2}$$

```
> combine(%);
```

$$\frac{\sinh(x) + I\sin(y)}{\cosh(x) + \cos(y)}$$